
//*********************************************
//
// NAME: WPCH_SR1.WPM
//
// PURPOSE: Search/Replace Coach
//
// AUTHOR: WordPerfect Corp. (G. McComb)
//
//*********************************************

// Revision History
//
//

// May 14, 1993 PAULL - Added hotspot capability for "Hint" and "Quit..Esc"

// May 13, 1993 PAULL - Changed the opening screen to match conventions.

Global(ScrnMode;MsgAdj;DefTitle;CancelMsg)
Global(HotElement;SaveScrn)

CancelMsg=0
select[0]=0
xpos=0
ypos=0

Deftitle="Coach"
ScrnMode=?ScreenHigh-24

StepNum=0 // Initialize step number
StepText="Position your cursor" // Initialize step text

Use("wpchlibr.wpm") // Shared library file
ColorScheme()
ScreenSetup()

MsgAdj=DlgMessage() // Set an adjustment for messages



//*********************************************
//
// MAIN PROGRAM: Coach
//
//*********************************************
Cancel(Off!)

Call(WelcomeDlg)
Call(PosCursor) // Call routine that helps position cursor
DisplayRewrite
WindowMaximize
Success=?Position // Set current cursor position to variable success 
Call(GetSRDlg) // help user Get to Replace dialog

Message(&tbox;1008)
Result=MessageBox("Coach";&tbox;6)

QuitCoach()

//*********************************************
// END MAIN PROGRAM
//*********************************************



//*******************************
//Subroutine:  Welcome
//********************************
Label(WelcomeDlg)

Message(&tbox;10)
Result=Welcome("Coach";&tbox)
If(Result=-1)

QuitCoach()
EndIf

Return



//***********************************************
//
//  Label:  PosCursor
//
//*****************************************************
Label(PosCursor)

Message(&tbox;50)
FakeIt(DefTitle;&tbox;3;1;0)
PosCharNext
PosCharPrev
keypress={-8154;}
chkcode={-1;}
Result=DoEdit(1;&chkcode;&keypress)

Return



//*********************************************
//
// SUBROUTINE: GetSRDlg
//
// PURPOSE:  Walk user through getting to the s&r dlg
//
//*********************************************
Label(GetSRDlg)

Message(&tbox;100)
hbox1={ "Choose [Bold On]Edit";

}
hbox2={ "Choose [Bold On]Replace";

}
hbox3={ -1;

}
EditMenu(&valid)
allow={ 1499; //replace menu option

}
exallow={ 0; // Allow execute

}
accel={ 108; //Rep(l)ace

}
exaccel={ 0; // Allow execute

}
nosubmenu(&subval;&subacl;&exsubacl;&subs)
funkey={ -8135 // Alt+F2  // -8135

}

SearchString("")
ReplaceString("")

Repeat
Result=PostIt(DefTitle;&tbox;38;11;0;2;0)
MenuActive=0
Show=HotSpot(0;0;62;0;-544;&funkey;1)
Switch(show)

CaseOf 1: // Show menu
sbox={"Choose [Bold On]Replace";}
Result=PostIt(DefTitle;&tbox;38;11;0;3;0)
ShowMenu("Coach";&sbox;37;16-ScrnMode;1;13;0;0)

CaseOf 1475: //Edit menu Id

CaseOf -544: // Accel menu  alt-e
PullDownDlg



Type("e")

CaseOf -8135: // Function key (Alt+F2)

CaseOf -8085: // Alt+=
sbox={-1;}
Result=PostIt(DefTitle;&tbox;38;11;0;3;0)
MenuActive=MenuBarControl("Coach";&hbox1;&sbox;37;
16-ScrnMode;101;&funkey)

DEFAULT: // Default
sbox={-1;}
MenuActive=MenuBarControl("Coach";&hbox1;&sbox;37;

16-ScrnMode;101;&funkey)
EndSwitch

If(MenuActive <> -1)
If(Show = -8135)

CancelMenus()
DisplayRewrite
ReplaceDlg
Call(ReplaceSub1)

Else
If(MenuActive = -8135)

Call(ReplaceSub1)
Else

Result=MenuControl(&hbox1;&hbox2;&hbox3;&valid;
&allow;&exallow;&accel;&exaccel;&subval;
&subacl;&exsubacl;&subs;&funkey)
MenuActive=Result
If(Result = 1499 or Result = 108)

CancelMenus()
DisplayRewrite
ReplaceDlg
Call(ReplaceSub1)

Else
If(Result = -8135)

Call(ReplaceSub1)
Endif

Endif
EndIf

Endif
EndIf

Until(MenuActive <> -1)
DlgInput(Off!)

Return



//***********************************************
//  LABEL:  ReplaceSub1
//
//  PURPOSE: Open and clear Replace dialog
//
//****************************************************
Label(ReplaceSub1)

Message(&tbox;105)
Result=PostIt("Coach";&tbox;10;1+(MsgAdj-ScrnMode);0;4;0)

//*********************************************
// The next command is a remedy for a bug
//*********************************************

PosCharNext
PosCharPrevious

//*********************************************
//*********************************************

While(?Dialog <> -1)
Char(Key)
If(?Mousey = (yposS1 + ScrnMode))

If((?Mousex >= xposH1) and (?Mousex <= xposH2))
ShowCode(SaveScreen!)
SpecialHint()
ShowCode(RestoreScreen!)

//*********************************************
// The next command is a remedy for a bug
//*********************************************

PosCharNext

//*********************************************
//*********************************************

Else
If((?Mousex >= xposQ1) and (?Mousex <= xposQ2))

CancelCoach()
Else

If(?DialogControl = 3543 or Key = -8159)
Display(Off!)
NotFound(On!)
OnNotFound(NotFoundError)



Endif
Type(ntoc(Key))

Endif
Endif

Else
If(Key = -8160)

CancelCoach()
Else

If(?DialogControl = 3543 or Key = -8159)
Display(Off!)
NotFound(On!)
OnNotFound(NotFoundError)

Endif
Type(ntoc(Key))

Endif
Endif
If(key = -8158) // F1

ShowCode(SaveScreen!)
SpecialHint()
ShowCode(RestoreScreen!)

//*********************************************
// The next command is a remedy for a bug
//*********************************************

PosCharNext

//*********************************************
//*********************************************

EndIf
EndWhile
Display(On!)
DisplayRewrite

Return



//*********************************************
//
// LABEL SpecialHint
//
//*********************************************
LABEL (SpecialHint)
Sty=StyNoPeriod!
Ln1=""
Ln2=""
Result=0
Repeat

OpOn={"";"";"";"";"";""}
OpOff={"";"";"";"";"";""}
If (Result>=1 And Result<=6)

OpOn[Result]="[Bold On]"
OpOff[Result]="[Bold Off]"

EndIf
DlgCreate(Result;"Hint";DlgNoClear!+DlgNoCancel!;;1;65;12)
DlgControl(CtrlOption!;;OpOn[1]+"Confirm Replacement"+OpOff[1];Sty;2;3)
DlgControl(CtrlOption!;;OpOn[2]+"Backward Search"+OpOff[2];Sty;2;4)
DlgControl(CtrlOption!;;OpOn[3]+"Case Sensitive Search"+OpOff[3];Sty;2;5)
DlgControl(CtrlOption!;;OpOn[4]+"File Whole Words Only"+OpOff[4];Sty;30;3)
DlgControl(CtrlOption!;;OpOn[5]+"Extended Search"+OpOff[5];Sty;30;4)
DlgControl(CtrlOption!;;OpOn[6]+"Limit Number of Matches"+OpOff[6];Sty;30;5)
DlgControl(CtrlLabel!;;"[Bold On]Choose a hint topic[Bold Off]";Sty;2;1)
DlgControl(CtrlLabel!;;Ln1;Sty;2;7)
DlgControl(CtrlLabel!;;Ln2;Sty;2;8)
DlgEnd
Switch(Result)

CaseOf 1: Ln1="Check this box if you want to skip over some replacements."
Ln2=""

CaseOf 2: Ln1="Check this box to change the search direction."
Ln2=""

CaseOf 3: Ln1="Check this box if, for example, you want to find all"
Ln2="occurrences of [Bold On]President[Bold Off], but not [Bold 
On]president[Bold Off]."

CaseOf 4: Ln1="Check this box if, for example, you want to find all"
Ln2="occurrences of [Bold On]apples[Bold Off], but not [Bold 
On]pineapples[Bold Off]."

CaseOf 5: Ln1="Check this box if you want to extend the search into"
Ln2="headers, footers, and footnotes."

CaseOf 6: Ln1="Check this box if you want to limit the number of"
Ln2="replacements."

EndSwitch
Until (Result=0)
Return



//*********************************************
//
// LABEL NotFoundError
//
//*********************************************
LABEL(NotFoundError)

Display(On!)
CancelDialogs()
DisplayRewrite
NotFoundCoach()
Message(&tbox;1007)
Result=MessageBox("Coach";&tbox;6)
QuitCoach()

Return



//*********************************************
//
// PROCEDURE: Message
//
//*********************************************
PROCEDURE Message(&tbox;Msg)

Switch(Msg)

CaseOf 10:
tbox={ "";

"            [Bold On]Using Search and Replace[Bold Off]";
"";
"This coach will help you use the Replace feature";
"to automatically search for and replace words";
"or phrases.";
"";

}

CaseOf 50:
tbox={ "";

"  Place your cursor at the top of your document.";▸
"";

}

CaseOf 100:
tbox={ "To search and replace text or codes,";

"";
"  Choose [Bold On]Edit[Bold Off],";▸
"  then [Bold On]Replace[Bold Off].";
"   or";
"  Press [Bold On]Alt+F2[Bold Off].";▸
"";

}

CaseOf 105:
tbox={ "";

"  Type the word or phrase you want to search for.";▸
"  Select [Bold On]Replace with:[Bold Off]";▸
"  Type the new word or phrase in the Replace With box.";▸
"  Select [Bold On]Replace";▸
"";

}

CaseOf 1007:
tbox={ "";



"The text you are searching for was not found.  Run the";
"coach again and make sure the text you entered into the";
"""Search For"" box was entered correctly.";
"";

}

CaseOf 1008:
tbox={ "";

"This concludes the coach on Search and Replace.";
"";

}

DEFAULT:
tbox={ "";

"Message not found!";
"";

}

EndSwitch

ENDPROC


